
CHAPTER 1

Formula basics and operators

A formula is an instruction that tells Excel to calculate something. Every formula
starts with an equals sign (=). If you type 5+3 in a cell, Excel just shows the text 5+3.
If you type =5+3, Excel shows the answer, 8.

Formulas become useful when they point at other cells instead of fixed numbers.
Each cell has an address made of its column letter and row number, such as B2. If
you type =B1+B2 and later change the number in B1, the answer updates by itself.
This is the whole magic of a spreadsheet.

How to enter a formula
1. Click the cell where you want the answer.
2. Type = and then the formula. You can click a cell instead of typing its address.

3. Press Enter. The cell shows the result; the formula bar above the grid shows the
formula itself.

To change a formula later, click the cell and edit it in the formula bar, or press F2.

The operators

An operator is a symbol that tells Excel what to do with two values.

Operator Meaning Example Result

+ Add =7+3 10

- Subtract =7-3 4

* Multiply =7*3 21

/ Divide =7/2 3.5

A Power =772 49

& Join text ="Pune"&"-411001" Pune-411001

Excel follows the BODMAS order you learnt in maths: brackets first, then powers,
then multiplication and division, then addition and subtraction. So =10+5*2 gives
20, but =(10+5)*2 gives 30.

Formulas 1 to 8 on a shop's day

Here is a small practice sheet. Rows 1to 5 come from Meera's stationery shop in
Nagpur; rows 6 to 9 are quick extra examples.

1 Opening cash (!) 5000
2 Sales today (!) 3250

3 Expenses today (!) 1180

4 Notebooks sold 24

5 Price of one notebook (!) 45

6 Restaurant bill (') 1360

7 Friends sharing the bill 4

8 Side of a square plot (m) 12

9 Ravi Kumar

1. Add. Cash after sales: =B1+B2 gives 8250.

2. Subtract. Cash at closing: =B1+B2-B3 gives 7070.
3. Multiply. Notebook sales: =B4*B5 gives 1080.

4. Divide. Each friend's share: =B6/B7 gives 340.

5. Percentage. Expenses as a part of sales: =B3/B2 gives 0.3630769. Click Percent
Style on the Home tab and it shows 36%. You can also write =B3/B2*100 to get
36.30769 as a plain number.

6. Power. Area of the plot: =B8~2 gives 144 square metres.

7. Join text. Full name: =A9&" "&B9 gives Ravi Kumar. The " " in the middle adds a
space.

8. Compare. Were sales more than expenses? =B2>B3 gives TRUE. The comparison
operators are =, >, <, >= (greater than or equal), <= and <> (not equal).

REMEMBER

Always start with =. A formula without it is just text, and Excel will not calculate it.

TIP

Type numbers without the ! sign or commas, such as 3250, and then format the
cells as currency. A typed "! 3,250" may be stored as text, which formulas cannot
add.

CHAPTER 2

Cell references: relative, absolute and mixed

When you copy a formula to another cell, Excel adjusts the cell addresses in it.
Understanding how it adjusts them is the most important skill in this book.

A B C D E
1 Item Qty Rate (!) Amount (1) After discount (!)
2 Shirt 2 800
3 Jeans 1 1500
4 Shoes 1 2200

Cell H1 holds the discount: 10%.

9. Relative reference. In D2 type =B2*C2.The result is 1600. Now drag the small
square at the bottom-right corner of D2 (the fill handle) down to D4. Excel writes
=B3*(3 in D3 and =B4*C4 in D4, giving 1500 and 2200. The addresses moved down
with the formula. This is called a relative reference, and it is what you want most of
the time.

10. Absolute reference. In E2 type =D2-D2*$H$1 . The dollar signs lock H1, so it
stays H1 when copied. Fill down and you get 1440, 1350 and 1980. Without the
dollar signs, E3 would look at H2, which is empty, and give the wrong answer. An
address with both dollar signs is an absolute reference.

11. Mixed reference. Sometimes you lock only the column or only the row. A
multiplication table shows why. Put 1to 5 in A2:A6 and 1to 5 in B1:F1.In B2 type
=$A2*B$1 and fill it across to F2 and down to F6. $A always points at column A, and
$1 always points at row 1. Cell D4 shows 9 (3 x 3) and F6 shows 25.

Written as Column when copied Row when copied
B2 changes changes
$BS2 fixed fixed
$B2 fixed changes
BS2 changes fixed
TIP

While typing a formula, click on an address and press F4. Excel adds the dollar
signs for you. Press F4 again to move through the four kinds shown in the table.

TRY THIS

Put a "rate per kg" of 60 in G1. List quantities in kg in A2:A6, and in B2 write a
formula for the cost that you can fill down. You should need $G$1 in it.

CHAPTER 3

Totals, averages and counts

These functions summarise a list. A function is a ready-made formula with a name,
such as SUM. The values you give it inside the brackets are called arguments. A
range like B2:B9 means every cell from B2 down to B9.

This is the maths test of a Class 9 section in Bhopal. Isha was absent, and Vikram's
mark has not been entered yet.

1 Student Maths
2 Aarav 78

3 Diya 92

4 Kabir 65

5 Isha Absent
6 Rohan 88

7 Sana 90

8 Vikram

9 Neha 54

12. SUM adds the numbers. =SUM(B2:B9) gives 467. It skips the text "Absent" and
the empty cell.

13. AVERAGE finds the mean. =AVERAGE (B2:B9) gives 77.83 (467 + 6). Text and
blank cells are left out, but a 0 would be counted.

14. MIN finds the smallest number. =MIN(B2:B9) gives 54.
15. MAX finds the largest number. =MAX(B2:B9) gives 92.
16. COUNT counts cells that hold numbers. =COUNT (B2:B9) gives 6.

17. COUNTA counts cells that are not empty, including text. =COUNTA (B2:B9) gives
7.

18. COUNTBLANK counts empty cells. =COUNTBLANK (B2:B9) gives 1, which reminds
the teacher that one mark is missing.

19. LARGE finds the second, third or any highest value. =LARGE (B2:B9,2) gives 90,
the second-highest mark.

20. SMALL works the same way from the bottom. =SMALL (B2:B9,2) gives 65.

Excel Formulas Workbook - Gurukul Technicals Institute 8



21. RANK gives a value's position in a list. =RANK(B6, $B$2:$B$9) gives 3 for Rohan,
because two students scored more. Lock the range with dollar signs before filling
down. If two values tie, they share a rank and the next rank is skipped. Newer
versions also offer RANK.EQ, which works the same way.

DID YOU KNOW

Select some numbers and look at the bar at the bottom of the Excel window. It
shows their Average, Count and Sum without any formula.

REMEMBER

COUNT counts numbers only, while COUNTA counts anything that is not empty. To
count students who appeared, use COUNT; to count names on a list, use COUNTA.

CHAPTER 4

Counting and adding with conditions

Often you do not want the total of everything, but the total of some things: only
Amit's sales, or only the rice. These functions take a criterion, a condition that
decides which cells count.

Sales register of a provision store in Kanpur:

A B C

1 Salesperson Product Sales (1)
2 Amit Rice 1200

3 Pooja Sugar 800

4 Amit Sugar 450

5 Ravi Rice 2000

6 Pooja Rice 1500

7 Amit Qil 950

8 Ravi Sugar 600

22. COUNTIF counts cells that meet one condition. =COUNTIF (A2:A8,"Amit") gives
3 sales by Amit. With numbers, put the comparison in quotes:
=COUNTIF(C2:C8,">=1000") gives 3 sales of ! 1,000 or more.

23. SUMIF adds the cells that match. =SUMIF(A2:A8,"Amit",C2:C8) gives 2600.
The first range is checked; the last range is added.

24. AVERAGEIF averages the matching cells. =AVERAGEIF (B2:B8, "Rice",(2:(C8)
gives 1566.67, the average rice sale.

25. COUNTIFS counts rows that meet several conditions at once.
=COUNTIFS(A2:A8,"Amit",B2:B8, "Sugar") gives 1.

26. SUMIFS adds with several conditions.
=SUMIFS(C2:C8,A2:A8,"Pooja",B2:B8,"Rice") gives 1500. Notice that in
SUMIFS the range to add comes first, while in SUMIF it comes last.

27. AVERAGEIFS averages with several conditions.
=AVERAGEIFS(C2:C8,B2:B8, "Sugar",C2:C8,">500") gives 700, the average of
the sugar sales above ! 500 (800 and 600).

Using a cell as the condition

Instead of typing "Amit" into the formula, put the name in a cell, say E1, and write
=SUMIF (A2:A8,E1,C2:C8) . Now you can type any name in ET and see that person's
total. For a comparison, join the operator to the cell with &:
=COUNTIF(C2:C8,">="&E2) .

TIP

Conditions are not case-sensitive. "amit”, "Amit" and "AMIT" all match. But an extra
space, as in "Amit ", does not match, so keep your data clean.

EXAMPLE

At the end of the month, the shop owner builds a small summary table: names in
E1:E3 and =SUMIF($A$2:$A$8,E1,$C$2:$C$8) in F1, filled down. Amit shows
2600, Pooja 2300 and Ravi 2600.

CHAPTER 5

Making decisions with IF

The IF function lets a sheet decide between two answers. It asks a question that is
either TRUE or FALSE, and gives one value for TRUE and another for FALSE.

The patternis: =IF(question, value if true, value if false).

A teacher in Amritsar has these results. In this example the pass mark is 33 and the
minimum attendance is 75%; your school's rules may be different.

A B C
1 Student Marks Attendance %
2 Tanvi 72 90
3 Arjun 28 80
4 Fatima 45 70
5 Harpreet 91 95

28.IF. =IF(B2>=33,"Pass","Fail") gives Pass for Tanvi. Filled down, Arjun gets
Fail. Text answers must be inside double quotes.

29. AND is TRUE only when every condition is true. =AND (B2>=33,C2>=75) gives
TRUE for Tanvi but FALSE for Fatima, whose attendance is 70.

30. OR is TRUE when at least one condition is true. A prize for marks of 90 or more
OR attendance of 95 or more: =0R (B5>=90,C5>=95) gives TRUE for Harpreet and
FALSE for Tanvi.

31. NOT reverses TRUE and FALSE. =NOT (B3>=33) gives TRUE for Arjun, because it is
true that he did not reach 33.

AND and OR are most useful inside IF: =IF (AND (B2>=33,(C2>=75), "Allowed", "Not
allowed") shows Allowed for Tanvi and Not allowed for Fatima.

32. Nested IF. To give more than two answers, put an IF inside another IF. For
grades: =IF(B2>=75,"A",IF(B2>=60,"B",IF(B2>=33,"C","D"))) . Tanvi (72) gets
B, Arjun D, Fatima C and Harpreet A. Excel checks from left to right and stops at the
first TRUE, so always start with the highest grade.

33. IFS does the same job with less bracket-counting:
=IFS(B2>=75,"A",B2>=60, "B",B2>=33,"C", TRUE, "D") . The last pair, TRUE and
"D", catches everything else. IFS needs Excel 2019 or later, or Microsoft 365.

34. IFERROR shows something friendlier when a formula fails. If A8 holds 1250 and
B8 holds 0, =A8/B8 shows the error #DIV/0!. =IFERROR (A8/B8, "Check") shows
Check instead.

REMEMBER

Every opening bracket needs a closing bracket. In a nested IF with three IFs, the
formula ends with three closing brackets. Excel colours each pair differently to
help you.

NOTE

Do not wrap every formula in IFERROR. It hides all errors, including the ones that
warn you about a real mistake. Use it only where you expect an error, such as a
lookup that may not find a value.

CHAPTER 6

Rounding and whole numbers

Money is usually shown to two decimal places, bills are often rounded to the
nearest rupee, and you cannot hire 2.6 buses. These functions handle such cases.

Cell A1 holds the number 1234.567.

35. ROUND rounds to the number of digits you choose. =ROUND (A1,2) gives
1234.57; =ROUND (A1,0) gives 1235; =ROUND (A1, -2) rounds to the nearest
hundred and gives 1200. A 5 rounds up,so =ROUND(2.5,0) gives 3.

36. ROUNDUP always rounds up. A school in Shimla is taking 130 students on a trip,
and each bus holds 50. =ROUNDUP (130/50,0) gives 3 buses, not 2.6.

37. ROUNDDOWN always rounds down. =ROUNDDOWN (A1,1) gives 1234.5.

38. INT drops the decimal part of a positive number. A sweet shop packs 45 laddoos
in boxes of 6. =INT(45/6) gives 7 full boxes.

39. MOD gives the remainder after division. =MOD (45,6) gives 3 laddoos left over.
MOD is also handy for spotting even numbers: =MOD (A2, 2) is O when A2 is even.

40. ABS gives the size of a number without its minus sign. A salesperson's target
was ! 5,000 and actual sales were ! 4,200. =ABS (4200-5000) gives 800, the gap,
whichever way round it is.

Formatting is not rounding
The Decrease Decimal button on the Home tab only changes how many decimals

you see. The full number stays in the cell and is used in later sums. ROUND actually
changes the value.

EXAMPLE

Three items cost ! 10.333 each after a discount. Shown with two decimals, each
looks like ! 10.33, which adds up to ! 30.99, yet SUM shows ! 31.00 because it adds
the hidden digits. Use =ROUND (price,2) on each item if the bill must add up
exactly as printed.

TRY THIS

A tailor in Surat needs 2.3 metres of cloth per kurta and has 20 metres. Use INT to
find how many kurtas she can make, and a formula to find the cloth left over.

CHAPTER 7

Working with text

Text functions clean up names, split codes and build sentences. They are a lifesaver
when data is copied from a website or typed by many people.

1 DEL-2026-0457

2 priya sharma

3 Priya.Sharma@Example.com
4 aMIT kumar

5 (two spaces) rahul (three spaces) verma (one space)

41. LEFT takes characters from the start. =LEFT (A1, 3) gives DEL, the city code.

42, RIGHT takes characters from the end. =RIGHT (A1,4) gives 0457, the serial
number.

43. MID takes characters from the middle: the text, the starting position and how
many characters. =MID(A1,5,4) gives 2026.

44. LEN counts characters, including spaces and hyphens. =LEN(A1) gives 13.It is
useful for checking that every mobile number in a list has 10 digits.

45. UPPER changes text to capitals. =UPPER(A2) gives PRIYA SHARMA.

46. LOWER changes text to small letters. =LOWER (A3) gives
priya.sharma@example.com, the usual way to store email addresses.

47. PROPER gives each word a capital first letter. =PROPER(A4) gives Amit Kumar.

48. TRIM removes extra spaces, keeping single spaces between words. =TRIM(A5)
gives rahul verma. Combine two functions to finish the job: =PROPER (TRIM(A5))
gives Rahul Verma.

49. CONCAT joins pieces of text. If B1 holds Neha and C1 holds Gupta, =CONCAT (B1, "
",C1l) gives Neha Gupta. CONCAT needs Excel 2019 or later, or Microsoft 365. Older
versions use CONCATENATE, which works the same way, and the & operator works
everywhere: =B1&" "&C1.

50. TEXT turns a number or date into text with a chosen format. ="Bill amount:
X"&TEXT (2450, "#,##0.00") gives Bill amount: ! 2,450.00. If D1 holds the date 15
August 2026, =TEXT (D1, "dd-mmm-yyyy") gives 15-Aug-2026.

TIP

After cleaning a column with formulas, you may want plain text without the
formulas. Copy the column, then use Paste Special and choose Values.

DID YOU KNOW

Without TEXT, joining a date to words shows a strange number. ="Due on "&D1
might show "Due on 46249", because Excel stores dates as numbers. TEXT fixes
that.

CHAPTER 8

Working with dates

Excel stores every date as a number: 1)anuary 1900 is 1,and each day adds 1. That is
why you can subtract one date from another to get the number of days between
them, or add 30 to a date to get a due date.

1 Date of birth 14-03-2010
2 Today's date for this example 01-10-2026
3 Invoice date 18-09-2026
4 Purchase date 15-01-2026

Dates here are written day-month-year, as in India. Your computer's region setting
decides how dates are typed and shown.

51. TODAY gives today's date: =TODAY () . It has no arguments, but the brackets are
still needed. It updates every time the file opens.

52. NOW gives today's date and the current time: =NOW() . Format the cell to show
the time.

Two simple date sums, with no function at all: the payment due date is =B3+30,
which gives 18-10-2026, and the days since the invoice are =B2-B3, which gives 13.

53. DATEDIF finds the gap between two dates in complete years, months or days.
=DATEDIF(B1,B2,"Y") gives 16,the age in full years. =DATEDIF(B1,B2,"YM")
gives 6, the extra months after those years. So the student is 16 years and 6 months
old. Use "M" for total months and "D" for days. Excel does not show DATEDIF in its
suggestion list, so type it in full; it still works.

54. EDATE gives the date a number of months later. A water purifier bought on
15-01-2026 has an 18-month warranty: =EDATE (B4, 18) gives 15-07-2027. If the cell
shows a number like 46583, format it as a date. EDATE handles short months
sensibly: one month after 31-01-2026 is 28-02-2026.

55. NETWORKDAYS counts working days from Monday to Friday, including both the
start and end dates. Put the dates in cells rather than typing them inside the
formula, because typed dates depend on the region setting. With 01-10-2026 in E1
and 31-10-2026 in E2, =NETWORKDAYS (E1,E2) gives 22. List the office's holidays in a
range, say G1:G2 with 02-10-2026 and 20-10-2026, and

=NETWORKDAYS (E1,E2,G1:G2) gives 20.

REMEMBER

TODAY changes every day. For a date that must stay fixed, such as a bill date, press
Ctrl+; to type today's date as a value instead.

EXAMPLE

Alibrary in Kochi lends books for 14 days. In the issue register, column C holds the
issue date and D holds =C2+14 as the return date. Column E shows
=IF (TODAY()>D2,"Late","") to flag late books.

CHAPTER 9

Looking up values

Lookup functions find a value in a table and bring back related information, like
finding a name in a phone book and reading the number beside it. Shop bills, fee
registers and stock sheets use them all the time.

Price list of a grocery shop in Jaipur (sheet cells A1:D6):

A B C D

1 Code ltem Unit Price (!)
2 P101 Basmati rice kg 120

3 P102 Toor dal kg 160

4 P103 Sugar kg 45

5 P104 Mustard oil litre 190

6 P105 Tea leaves 250g 140

56. VLOOKUP searches down the first column of a table and returns a value from
another column in the same row. =VLOOKUP ("P103", $A$2:$D$6,4,FALSE) gives
45.The parts are: what to find, where to look, which column to return (column 4 of
the table) and FALSE for an exact match. Leaving out FALSE can give wrong answers,
so always include it for codes and names.

57. HLOOKUP does the same across a row. Monthly sales are in A9:E10:

A B C D E
9 Month Jan Feb Mar Apr
10 Sales (!) 42000 38500 51200 47800

=HLOOKUP ("Mar",B9:E10,2,FALSE) gives 51200.

58. INDEX returns the value at a given position in a range. =INDEX(B2:B6,4) gives
Mustard oil, the fourth item. =INDEX(A2:D6,2,4) gives 160: row 2, column 4 of
the table.

59. MATCH finds the position of a value in a list. =MATCH("Sugar",B2:B6,0) gives
3. The O means exact match.

Together, INDEX and MATCH can look in any direction. VLOOKUP cannot return a
column to the left of the one it searches, but this can:
=INDEX(A2:A6,MATCH("Sugar",B2:B6,0)) gives P103, the code for sugar.

60. XLOOKUP is the modern lookup. You give it what to find, the column to search,
the column to return and, if you like, a message for when nothing is found.
=XLOOKUP ("Tea leaves",B2:B6,D2:D6,"Not found") gives 140, and the same
formula with "Ghee" gives Not found. It looks left or right and matches exactly by
default. XLOOKUP needs Excel 2021 or Microsoft 365; in older versions it shows
#NAME?, so use VLOOKUP or INDEX and MATCH there.

TIP

Lock the table range with dollar signs ($A$2:5D$6) before you fill a lookup formula
down. Otherwise the range slides down with each row and stops finding the first
items.

DID YOU KNOW

Typing codes instead of item names on a bill avoids spelling mistakes. One letter
wrong in "Basmati rice" and the lookup fails; a short code is easier to type
correctly.

CHAPTER 10

Three projects: marksheet, shop bill and
salary sheet

These projects combine the formulas you have learnt. Build each one yourself;
typing the formulas teaches more than reading them.

Project 1: a class marksheet

Columns: A Name, B Hindi, C English, D Maths, E Science, F Social Science (each out

of 100). Cell O1 holds the pass mark for each subject; we use 33 as an example.

Eng Maths i % Result Grade Rank
Ananya 78 85 92 88 81 424 84.8%  Pass A 2
Mohit 65 58 30 61 70 284 56.8% Fail C 4
Sneha 90 88 95 93 89 455 91.0%  Pass A 1

Name Hin Eng Maths Sci SSc Total % Result Grade Rank

Imran 72 69 75 80 66 362 72.4%  Pass B 3

The formulas in row 2, filled down to row 5:

1. Total in G2: =SUM(B2:F2)
2. Percentage in H2: =G2/500, formatted as a percentage.

3.Resultin 12: =IF(COUNTIF(B2:F2,"<"&$0$1)=0, "Pass","Fail") . It counts
subjects below the pass mark; if there are none, the student passes. Mohit fails
because of his 30 in Maths, even though his total is fine.

4. Grade in J2: =IF(H2>=75%, "A", IF (H2>=60%, "B", IF (H2>=33%,"C","D")))
5.Rank in K2: =RANK (G2, $6$2:$G$5)
Below the table, add a summary: class average =AVERAGE (G2:G5) gives 381.25;

highest total =MAX(G2:G5) gives 455; students passed =COUNTIF(I2:I5,"Pass")
gives 3; and the topper's name =INDEX(A2:A5,MATCH(MAX(G2:G5),G2:G5,0))

gives Sneha.

Project 2: a shop bill

B C D E
1 Item Qty Rate (!) Amount (!)
2 Notebook 6 55 330
3 Ball pen pack 2 50 100
4 Geometry box 1 120 120
5 Drawing file 3 35 105

1. Amount in E2, filled down: =C2*D2

2. Subtotal in E6: =SUM(E2:E5) gives 655.

3. A 5% discount on bills of | 500 or more, in E7: =IF (E6>=500,E6*5%,0) gives
32.75.

4., After discount, in E8: =E6-E7 gives 622.25.

5. Bill amount rounded to the rupee, in E9: =ROUND (E8,0) gives 622.

6. Number of items: =COUNTA(B2:B5) gives 4; total pieces: =SUM(C2:C5) gives 12.

To fill rates automatically, keep the price list on a sheet named Prices and use
=VLOOKUP (B2,Prices!$A$2:$B$100,2,FALSE) in D2.The exclamation mark
separates the sheet name from the range.

NOTE

If your business charges GST, the rate depends on the item and the rules change
from time to time. Keep each rate in its own cell, never inside a formula, and check
current rates on gst.gov.in or with your accountant.

Project 3: a monthly salary sheet

All figures are made up for practice. Statutory deductions such as provident fund
have their own rules, so they are left out; ask an accountant before adding them to

a real sheet.

DEWH Earned oT oT
. Allowance Gross Advance
present basic hours pay
Kiran 18000 30 28 16800 2000 5 500 19300 1000
Deepak 15000 30 30 15000 1500 0 0 16500 O
Lata 21000 30 26 18200 2500 8 800 21500 2000

With the overtime rate per hour (! 100 in this example) in M1, the formulas in row 2
are:

1. Earned basic in E2: =ROUND (B2/C2*D2,0) . Rounding matters in a 31-day month:
15000 =+ 31x 29 is 14032.26, which rounds to 14032.

2. Overtime pay in H2: =G2*$M$1

3. Gross in 12: =E2+F2+H2

4. Net pay in K2: =12-32

5. Total net pay: =SUM(K2:K4) gives 54300.

TRY THIS

Add a column that shows "Full attendance" when days present equals days in

nn

month, and nothing otherwise. Hint: an empty answer is written as "".

CHAPTER 11

Fixing common errors

Every Excel user sees error messages. They are not a disaster; each one tells you

what kind of mistake to look for.

What it .
Common cause How to fix it
means
#DIV/O!  Division by Dividing by an empty  Fill in the missing number, or use
zero celloro =IF(B2=0,"",A2/B2)
#N/A Not available A lookup could not Check spelling and extra spaces; use
find the value IFERROR or XLOOKUP's "not found"
message

#VALUE!  Wrong kind Doing maths on text, ~ Keep units in the heading, numbers

of value such as "12 kg"*45 in the cells
#REF! Invalid A cell the formula Undo with Ctrl+Z, or rewrite the
reference used was deleted formula
#NAME?  Name not A misspelt function, Check spelling; put text inside
recognised or text without double quotes
quotes
HHHH#H# Not an error The column is too Widen the column

narrow to show the
number or date

Worked examples

» =VLOOKUP("P109", $A$2:$D$6,4,FALSE) gives #N/A, because there is no item
P109 in the price list.

« =SUMM(B2:B9) gives #NAME?, because SUMM is not a function.
« =IF(B2>=33,Pass,Fail) also gives #NAME?, because Pass and Fail need
quotes.

« If D2 holds =B2*C2 and you delete column C, the formula turns into =B2*#REF ! .

Finding the cause

« Click the cell and read the formula in the formula bar. Excel colours each address
and draws a box of the same colour around that cell, so you can see what the
formula uses.

« On the Formulas tab, Evaluate Formula shows the calculation one step at a
time, and Trace Precedents draws arrows to the cells the formula depends on.

A small green triangle in the corner of a cell is a warning, not an error. It often
means a number is stored as text. Click the cell and the warning button that
appears explains it.

TIP

Press Ctrl+ (the key left of 1) to see every formula on the sheet instead of the
results. It is the quickest way to check a whole sheet. Press it again to go back.

REMEMBER

If your computer's region setting uses a semicolon instead of a comma between
arguments, write =ROUND (A1;2) instead of =ROUND (A1, 2) . Excel shows the right
separator in the tip box as you type.

CHECK YOURSELF

Practice questions

1. What must every Excel formula start with?
2. What results do =10+5*2 and =(10+5)*2 give, and why are they different?

3. Cell B3 holds 1180 and B2 holds 3250. Write a formula for B3 as a percentage of
B2, and say how you show it as a percentage.

10.

11.

12.

13.

14.

15.

16.

17.

18.

19.

20.

21.

22.

. What is the difference between B2, $B$2 and B$2 when a formula is copied?
. Which key adds dollar signs to a cell address while you type a formula?

. Arange holds 78,92, "Absent", an empty cell and 54. What do COUNT and

COUNTA give?

. Which function finds the second-highest value in a list?

. Write a formula to add the sales in C2:C8 made by "Pooja", whose names are in

A2:A8.

. In SUMIFS, which range comes first: the range to add or the range to check?

Write a formula to count the cells in C2:C8 that are ! 1,000 or more.

Write an IF formula that shows "Pass" when B2 is 33 or more and "Fail"
otherwise.

What does =AND(B2>=33,(2>=75) give when B2 is 45 and C2 is 70?
Why should a nested IF for grades start with the highest grade?

130 students are going on a trip and each bus holds 50. Which formula gives the
number of buses?

45 laddoos are packed in boxes of 6. Which two functions give the full boxes and
the leftover laddoos?

Cell A1 holds DEL-2026-0457. Write formulas to get "DEL" and "2026".

A name has been typed as " rahul verma ". Which formula turns it into "Rahul
Verma"?

What does =DATEDIF(B1,B2,"Y") find?

A product bought on 15-01-2026 has an 18-month warranty. Which function
gives the end date, and what is the date?

In =VLOOKUP(G1, $A$2:$D$6,4,FALSE) ,what do the 4 and the FALSE mean?
Why can INDEX and MATCH do something VLOOKUP cannot?

A colleague's XLOOKUP formula shows #NAME? on an older computer. What is
the likely reason?

23. What does #DIV/0! mean, and how can you avoid it?

24. A cell shows ###4##. Is this an error, and what should you do?

ANSWER KEY

Answers

10.
11.
12.

. An equals sign (=). Without it, Excel

treats the entry as text.

. 20 and 30. Excel multiplies before it

adds, but brackets are worked out first.

. =B3/B2, then click Percent Style on the

Home tab. It shows 36%.

. B2 changes both column and row; $B$2

never changes; BS2 lets the column
change but keeps row 2 fixed.

. F4.

. COUNT gives 3 (numbers only) and

COUNTA gives 4 (every cell that is not
empty).

. LARGE, for example =LARGE (B2:B9,2) .
. =SUMIF (A2:A8,"Pooja",C2:(C8).

. The range to add comes first in SUMIFS,

followed by pairs of a range and a
condition.

=COUNTIF(C2:C8,">=1000").
=IF(B2>=33, "Pass","Fail").

FALSE, because the second condition
(attendance of at least 75) is not met.

13.

15.

16.

17.

18.

21.

Excel stops at the first condition that is
TRUE. If a lower grade came first, a high
mark such as 91 would also pass that
test and get the lower grade.

. =ROUNDUP (130/50,0) , which gives 3.

=INT(45/6) gives 7 full boxes and
=MOD(45,6) gives 3 left over.

=LEFT(A1,3) gives DEL and
=MID(A1,5,4) gives 2026.

=PROPER(TRIM(A5)) ,where A5 holds
the name. TRIM removes the extra
spaces and PROPER fixes the capitals.

The number of complete years between
the two dates, such as a person's age.

. EDATE: =EDATE(B4,18) gives

15-07-2027 (format the cell as a date).

. 4 means return the value from the

fourth column of the table; FALSE means
find an exact match only.

VLOOKUP can only return columns to
the right of the column it searches.
INDEX and MATCH can return a value
from any column, including one on the
left.

22. XLOOKUP needs Excel 2021 or Microsoft 23. It means a number is being divided by

365. Older versions do not know it, so
they show #NAME?. Use VLOOKUP or
INDEX and MATCH instead.

WORD LIST

Glossary

zero or by an empty cell. Fill in the
missing value, or use a formula such as
=IF(B2=0,"",A2/B2).

24. No, it is not an error. The column is too

narrow to show the value; widen it.

Formula
an instruction that starts with = and tells
Excel to calculate something.

Function
a ready-made formula with a name, such
as SUM or IF.

Argument
a value or range you give a function inside
its brackets.

Cell address
the column letter and row number that
name a cell, such as B2.

Range
a block of cells written with a colon, such
as B2:B9.

Operator
a symbol such as +,-,* / or & that tells
Excel what to do with values.

Formula bar
the strip above the grid that shows the
formula in the selected cell.

Fill handle

the small square at the bottom-right
corner of a selected cell, dragged to copy
a formula.

Relative reference
a cell address that changes when the
formula is copied, such as B2.

Absolute reference

a cell address locked with dollar signs,
such as $B$2, that does not change when
copied.

Mixed reference
an address with only the column or only
the row locked, such as $B2 or BS$2.

Criterion

a condition, such as ">=1000" or "Amit",
that decides which cells are counted or
added.

Nested IF
an IF function placed inside another IF to
choose between more than two answers.

Logical value
the answer TRUE or FALSE.

Lookup

finding a value in a table and returning
related information from the same row or
column.

Exact match
a lookup setting that accepts only a value
that is exactly the same.

Serial number (date)
the number Excel uses to store a date,
counting days from 1 January 1900.

Working days
Monday to Friday, as counted by
NETWORKDAYS, less any listed holidays.

Error value
a message such as #N/A or #DIV/0! that
shows a formula cannot give an answer.

Paste Special (Values)
pasting only the results of formulas,
without the formulas themselves.

